
ПРЕДИСЛОВИЕ

Настоящее учебное пособие посвящено важному раз-
делу программирования – обработке данных, представ-
ленных в виде таблиц. Предполагается, что к началу
изучения этого раздела учащиеся приобрели навыки со-
здания простых программ (умеют использовать присва-
ивания, условные операторы, операторы цикла, а также
знакомы с описаниями и вызовами процедур и функций).

Учитывая, что наши читатели – начинающие програм-
мисты, мы постарались как можно подробнее комменти-
ровать разбираемые 3примеры программ, поясняя иногда
и те конструкции, которые изучались ранее, но могут вы-
звать определенные трудности при чтении пособия.

ДВУМЕРНЫЕ МАССИВЫ.
ОСОБЕННОСТИ ВВОДА

Одним из удобных способов представления и хранения
данных являются таблицы – страницы школьных журна-
лов, платежные ведомости, метеосводки и многое другое.
Наша цель – научиться обрабатывать данные, представ-
ленные в табличной форме, а именно, научиться вводить
их в память компьютера, выводить на экран и распеча-
тывать в виде таблиц различного формата, производить
манипуляции со строками, столбцами и отдельными эле-
ментами и, наконец, познакомиться с простыми, но важ-
ными алгоритмами обработки таблиц – матричными вы-
числениями.

Для представления таблиц в программе используются
двумерные массивы. Двумерный массив – это набор эле-
ментов одного и того же типа, упорядоченный (прону-
мерованный) с помощью двух индексов: первый индекс
соответствует номеру строки, а второй – номеру столбца,
на пересечении которых стоит этот элемент.

Рассмотрим таблицу A, состоящую из трех строк и че-
тырех столбцов (для краткости в дальнейшем будем гово-
рить, что такая таблица имеет размер 3 на 4), элементами
которой являются вещественные числа:

a11 a12 a13 a14
a21 a22 a23 a24
a31 a32 a33 a34

Указать, какую именно память следует отвести для
хранения значений элементов таблицы, следует, как обычно,
в разделе описаний. В нашем случае описание двумерного
массива будет выглядеть так:

var a: array [1..3, 1..4] of real

Из описания видно, что а – это идентификатор пере-
менной, представляющий двумерный массив, номера строк
которого могут принимать значения 1, 2, 3, а номера столб-
цов – 1, 2, 3, 4. Элементами такого массива должны быть
вещественные числа. Чтобы обратиться к отдельному
элементу массива, нужно указать, какое место в таблице
он занимает, например

a[1,1], a[1,3], a[2,4], a[3,3]

При этом первый индекс – это номер строки, а второй
– номер столбца, т.е. a[2,4] – элемент, стоящий на пересе-
чении второй строки и четвертого столбца. Обращения к
элементам a[4,3] или a[3,5] ошибочны, так как в первом
случае номер строки, а во втором – номер столбца при-
нимают недопустимые значения, вылезающие за границы
диапазонов, указанных в описании массива.

ПРИМЕР 1. Дана таблица из 4 строк и 5 столб-
цов, элементами которой являются вещественные числа.
Найти сумму всех элементов этой таблицы.

program SumOfElements;
var a: array [1..4, 1..5] of real;

sum: real;
i, j: integer;

begin
writeln; writeln (’Введи 20 чисел’);
for i := 1 to 4 do
for j := 1 to 5 do
begin
write (’a[’, i, ’,’, j, ’] = ’);
readln (a[i,j])

end;
sum := 0;
for i := 1 to 4 do
for j := 1 to 5 do
sum := sum + a[i,j];

writeln(’Сумма элементов таблицы равна ’,
sum:8:3)

end.

Раздел описаний этой программы состоит из описания
двумерного вещественного массива а (эта область памяти
будет использована для хранения элементов таблицы), ве-
щественной переменной sum (суммы элементов) и пере-
менных целого типа i и j.

В начале операторной части программы после опера-
тора перевода курсора на новую строку экрана и под-
сказки-напоминания программисту ’Введи 20 чисел’ с
помощью двух вложенных операторов циклов осуществ-
ляется ввод с клавиатуры элементов таблицы. Разберем
работу этой конструкции. Параметр внешнего цикла i
последовательно принимает значения от 1 до 4 (1, 2, 3,
4), и при каждом фиксированном значении i параметр j
пробегает значения от 1 до 5 шагом 1. Таким образом,
первые пять чисел, введенные с клавиатуры, будут соот-
ветствовать элементам массива

a[1,1], a[1,2], a[1,3], a[1,4], a[1,5]

то есть определят значения элементов первой строки таб-
лицы. Следующие введенные пять чисел (при значении
параметра i=2 и параметра j, принимающего последова-
тельно значения 1, 2, 3, 4, 5) определят элементы второй
строки таблицы и так далее, то есть заполнение таблицы
произойдет по строкам.

Следующий оператор присваивания определяет началь-
ное значение переменной sum, равное нулю, а вложенные
циклы с параметрами i и j подсчитывают сумму элемен-
тов массива. Отметим, что в этом варианте программы
суммирование элементов таблицы осуществляется по стро-
кам. Завершает программу вывод результата на экран.



7

Используемый при этом формат для вывода веществен-
ных чисел обеспечивает представление результата в при-
вычной форме. Отметим, что по умолчанию была бы ис-
пользована экспоненциальная форма записи веществен-
ных чисел.

Применимость этой программы весьма ограничена: а
именно, она пригодна для работы с таблицами из четы-
рех строк и пяти столбцов. При обработке таблиц раз-
мера m на n ( m строк и n столбцов) можно поступить
так: описать целые константы m и n и определить их
значения. Далее использовать эти константы в описании
двумерного массива, а затем во вложенном цикле в каче-
стве наибольших значений, принимаемых параметрами i
и j соответственно. Приведем этот вариант программы:

program SumOfElements1;
const m=4; n=5;
var a: array [1..m,1..n] of real;

sum: real;
i, j: integer;

begin
writeln; writeln (’Введи ’, m*n, ’ чисел’);
for i := 1 to m do
for j := 1 to n do
begin

write (’a[’, i, ’,’, j, ’] = ’);
readln (a[i,j])

end;
sum := 0;
for i := 1 to m do
for j := 1 to n do

sum := sum + a[i,j];
writeln(’Сумма элементов таблицы равна ’,

sum:8:3)
end.

В этом варианте программы изменение размеров обра-
батываемой таблицы потребует изменения значений кон-
стант m и n, после чего, естественно, текст программы
должен быть заново откомпилирован.

В тех случаях, когда нужно многократно исполнять
программу, обрабатывая таблицы, размеры которых же-
лательно вводить с экрана в каждом конкретном случае,
можно намеренно увеличить значения констант, опреде-
ляющих размеры массива, захватив при описании мас-
сива заведомо большую область памяти. Затем вводятся
значения переменных, определяющих истинные размеры
таблицы, и значения самих элементов, которые заполнят
только нужную часть отведенной области памяти.

program SumOfElements2;
const mmax = 20; nmax = 25;
var a: array [1..mmax, 1..nmax] of real;

sum: real;
i, j, m, n: integer;

begin
writeln;
write(’Введи количество строк ’);
readln(m);
write(’Введи количество столбцов ’);
readln(n);
writeln (’Введи ’, m*n, ’ чисел’);

for i := 1 to m do
for j := 1 to n do
begin

write (’a[’, i, ’,’, j, ’] = ’);
readln (a[i,j])

end;
sum := 0;
for i := 1 to m do
for j := 1 to n do
sum := sum + a[i,j];

writeln(’Сумма элементов таблицы равна ’,
sum:8:3)

end.

Используя такой способ, нужно соблюдать осторож-
ность, так как, во-первых, можно превысить допустимые
объемы памяти, а, во-вторых, будет затруднена диагно-
стика ошибок, связанных с контролем значений индексов
элементов массива.

ОСНОВНЫЕ ПРОЦЕДУРЫ
ОБРАБОТКИ ТАБЛИЦ

В этом разделе мы научимся оформлять отдельные ал-
горитмы, обрабатывающие таблицы, в виде процедур и
функций (описывать процедуры и функции) и использо-
вать (вызывать) их в самой программе.

ПРИМЕР 2. Даны две целочисленные таблицы A
и B размера m на n ( m строк и n столбцов в каждой).
Построить третью таблицу C, каждый элемент которой

равен сумме соответствующих элементов таблиц A и B ,
то есть cij = aij + bij при i = 1; : : : ;m и j = 1; : : : ; n. Все
таблицы распечатать.

Для решения этой задачи опишем четыре процедуры:
три процедуры, которые обеспечивают ввод таблиц, их
вывод и поэлементное сложение (параметрами этих про-
цедур будут двумерные массивы) и процедуру, с помощью
которой введем истинные размеры таблиц.

В начале программы опишем константыmmax и nmax,
определяющие максимальные размеры обрабатываемых
таблиц. Затем опишем новый тип table, который будем
использовать при спецификации параметров процедур:

const mmax = 20; nmax = 30;
type table = array [1..mmax, 1..nmax] of integer

В раздел описания переменных поместим описания трех
переменных a, b, c типа table и две переменные m и n
типа integer (настоящие размеры обрабатываемых таб-
лиц).

var a, b, c: table;
m, n: integer

Перейдем к описанию процедур. Первой опишем про-
цедуру, устанавливающую размеры таблиц:

procedure SizeOfTable;
begin
writeln;
write (’Введи количество строк ’);
readln (m);

write (’Введи количество столбцов ’);
readln (n);
writeln(’Программа работает с таблицами ’,

’ размера ’, m, ’ на ’, n, ’ !!! ’)
end

Это описание процедуры без параметров. В теле проце-
дуры глобальные переменные m и n получают значения,
введенные с клавиатуры.

Следующим сделаем описание процедуры, с помощью
которой будем с клавиатуры вводить элементы таблиц:

procedure InputTable (var x: table);
var i, j: integer;
begin
writeln;
writeln (’Введи ’, m*n, ’ чисел’);
for i := 1 to m do
for j := 1 to n do
begin write (’x[’, i, ’,’, j, ’]=’ );

readln (x[i,j])
end

end

Алгоритм, описанный в теле этой процедуры, был по-
дробно разобран в предыдущем разделе. Отметим, что
формальный параметр x должен быть специфицирован
как параметр-переменная (элементы массива x получают
значения, вводимые с клавиатуры).

Приступим к описанию процедуры, выводящей значе-
ния элементов таблицы на экран. Естественно потребо-



12

вать, чтобы значения выводились в табличной форме с
соблюдением строк и столбцов. Кроме того, предусмот-
рим наличие разделителя между таблицами.

procedure OutputTable (x: table);
var i, j: integer;
begin
writeln;
writeln (’**** Т А Б Л И Ц А ****’);
writeln;
for i := 1 to m do
begin
for j := 1 to n do write (x[i,j]:5, ’ ’ );
writeln

end
end

Последняя процедура, которую нам нужно описать,
строит новую таблицу, складывая две данных таблицы
поэлементно. Зададимся формальными параметрами x,
y, z, причем первые два будут соответствовать слагае-
мым, а третий – результату.

procedure Add (x, y: table; var z: table);
var i, j: integer;
begin
for i := 1 to m do
for j := 1 to n do
z[i,j] := x[i,j] + y[i,j]

end

Приведем полный текст этой программы:

program addition;
const mmax = 20; nmax = 30;
type table = array [1..mmax, 1..nmax] of integer;
var a, b, c: table;

m, n: integer;

procedure SizeOfTable;
begin
writeln;
write (’введи количество строк ’);
readln (m);
write (’введи количество столбцов ’);
readln (n);
writeln;
writeln (’Программа работает ’ ,

’ с таблицами размера ’ ,
m, ’ на ’, n, ’ !!! ’)

end;

procedure InputTable (var x: table);
var i, j: integer;
begin
writeln;
writeln (’Введи ’, m*n, ’ чисел’);
for i := 1 to m do
for j := 1 to n do
begin write (’x[’, i, ’,’, j, ’]=’ );

readln (x[i,j])
end

end;

procedure OutputTable (x: table);
var i, j: integer;
begin
writeln;
writeln (’**** Т А Б Л И Ц А ****’);
writeln;
for i := 1 to m do
begin
for j := 1 to n do

write (x[i,j]:5, ’ ’ );
writeln

end
end;

procedure Add (x, y: table; var z: table);
var i, j: integer;
begin
for i := 1 to m do
for j := 1 to n do
z[i,j] := x[i,j] + y[i,j]

end;
begin SizeOfTable;

InputTable (a);
InputTable (b);
OutputTable (a);
OutputTable (b);
Add (a, b, c);
OutputTable (c)

end.

Введем несколько определений. Во многих разделах
математики прямоугольные таблицы, состоящие из m

строк и n столбцов, то есть размера (m×n), принято на-
зывать матрицами (от латинского matrix) размера (m×n).
При m=n матрицу называют квадратной. Следом квад-
ратной матрицы называют сумму элементов главной диа-
гонали – диагонали, идущей от левого верхнего угла мат-
рицы в правый нижний угол.

Матрицы A и B считают равными, если они одинако-
вого размера (число строк и число столбцов матрицы A
соответственно равны числам строк и столбцов матрицы
B) и элементы, стоящие в матрице A и матрице B на оди-
наковых местах, равны между собой:

aij = bij (i = 1; :::;m; j = 1; :::; n):

ПРИМЕР 3. Даны две квадратных матрицы A
и B размера (m×n), элементами которых являются веще-
ственные числа. Определить матрицу, имеющую больший
след.

Для решения этой задачи опишем процедуры, связан-
ные с вводом и выводом матриц, аналогичные процеду-
рам из примера 2, а также новые процедуры и функции.
В начале программы опишем константуmmax, определя-
ющую максимальный размер обрабатываемых матриц, и
тип matrix (он будет использоваться при спецификации
параметров процедур и функций):

const mmax = 25;
type matrix = array [1..mmax, 1..mmax] of real

Подсчет суммы элементов главной диагонали оформим
в виде функции с заголовком

functuon Trace (x: matrix): real

то есть функции с одним параметром, выдающей веще-
ственное значение. Приведем описание функции Trace:

function Trace (x: matrix): real;
var i: integer;

sum: real;
begin
sum := 0;
for i := 1 to m do
sum := sum + x[i,i];

Trace := sum
end

Так как у элементов главной диагонали номера строк сов-
падают с номерами столбцов, то для их суммирования
достаточно одного цикла. Напомним также, что в теле
описываемой функции непременно должен быть оператор
присваивания, в левой части которого стоит идентифика-
тор функции, а в правой – выражение, значение которого
и определяет выдаваемое этой функцией значение. Пол-
ный текст программы будет выглядеть так:

program MaxTrace;
const mmax = 25;
type matrix = array [1..mmax, 1..mmax] of real;
var a, b: matrix;

m: integer;
procedure SizeOfMatrix;
begin
writeln;
write (’введи размер матриц ’);

readln (m)
end;

procedure InputMatrix (var x: matrix);
var i, j: integer;
begin
writeln;
writeln (’Введи ’, m*m, ’ чисел’);
for i := 1 to m do
for j := 1 to m do
read (x[i,j])

end;
procedure OutputMatrix (x: matrix);
var i, j: integer;
begin
writeln;
writeln (’***** М А Т Р И Ц А *****’);
writeln;
for i := 1 to m do
begin
for j := 1 to m do
write (x[i,j]:5, ’ ’ );

writeln
end

end;
function Trace (x: matrix): real;
var i: integer;
sum: real;
begin
sum := 0;
for i := 1 to m do
sum := sum + x[i,i];

Trace := sum
end;

procedure Comparison (a, b: matrix);
begin
writeln;
if Trace (a) > Trace (b)
then writeln (’След матрицы A ’,

’ больше следа матрицы B ’)
else if Trace (a) < Trace (b)

then writeln (’След матрицы B ’,
’ больше следа матрицы A ’)

else writeln (’Следы матриц равны ’)
end;

begin
SizeOfMatrix;
InputMatrix (a);
InputMatrix (b);
OutputMatrix (a);
OutputMatrix (b);
Comparison (a, b)

end.

Обратите внимание на то, что функция Trace является
вспомогательной для процедуры Comparision: вызовы
функции Trace(A) и Trace(B) находятся в теле проце-
дуры Comparision и исполняются при вызове этой про-
цедуры.

ПРИМЕР 4. Описать процедуру, меняющую ме-
стами элементы квадратной матрицы A, стоящие сим-
метрично относительно главной диагонали, иначе говоря,

транспонирующую квадратную матрицу. Элементами мат-
рицы являются вещественные числа. Матрица имеет раз-
мер (m×m).

В качестве примера рассмотрим матрицу A, размера
4×4, элементы которой обозначим aij :

a11 a12 a13 a14
a21 a22 a23 a24
a31 a32 a33 a34
a41 a42 a43 a44

Очевидно, что элементы главной диагонали (у них но-
мер строки i совпадает с номером столбца j) должны остать-
ся на своих местах. Переберем по строкам элементы, стоя-
щие выше главной диагонали, и поменяем их значения со
значениями соответствующих элементов, стоящих ниже
главной диагонали. Номера строк при этом пробегут зна-
чения 1; : : : ;m-1 и при каждом фиксированном значении
i номера столбцов j должны пробежать значения от i+1
до m. Приведем описание процедуры:

procedure TransposeOfMatrix (var a: matrix);
var i, j: integer;

temp: real;
begin
for i:= 1 to m-1 do
for j := i+1 to m do
begin
temp:=a[i,j]; a[i,j]:=a[j,i]; a[j,i]:=temp

end
end

Проверим работу этой процедуры. Воспользуемся тек-
стом программы из примера 4: удалим описания ненуж-
ные процедур и дополним этот раздел описанием проце-
дуры TransposeOfMatrix. Подкорректируем также ос-
новной текст программы:

program Transpose;
const mmax = 25;
type matrix = array [1..mmax, 1..mmax] of real;
var a: matrix;

m: integer;
procedure SizeOfMatrix;
begin
writeln;
write (’введи размер матриц ’);
readln (m)

end;
procedure InputMatrix (var x: matrix);
var i, j: integer;
begin
writeln;
writeln (’Введи ’, m*m, ’ чисел’);
for i := 1 to m do

for j := 1 to m do
read (x[i,j])

end;
procedure OutputMatrix (x: matrix);
var i, j: integer;
begin
writeln;
writeln (’****** М А Т Р И Ц А ******’);

writeln;
for i := 1 to m do
begin
for j := 1 to m do
write (x[i,j]:5, ’ ’ );

writeln
end

end;
procedure TransposeOfMatrix (var a: matrix);
var i, j: integer;

temp: real;
begin
for i:= 1 to m-1 do
for j := i+1 to m do
begin
temp:=a[i,j]; a[i,j]:=a[j,i]; a[j,i]:=temp

end
end;
begin
SizeOfMatrix;
InputMatrix (a);
OutputMatrix (a);
TransposeOfMatrix (a);
OutputMatrix (a)

end.

Определим еще одну важную операцию над матрицами
– умножение матриц. Произведением матрицы A размера
(m×n) на матрицу B размера (n×p) называется матрица
C размера (m×p), элемент cij которой (то есть элемент,

стоящий в i-й строке и j-м столбце) есть сумма произведе-
ний элементов i-й строки матрицы A на соответствующие
элементы j-го столбца матрицы B, то есть

cij =
n∑

k=1

aik bkj

Умножение определяется только для такой пары матриц
A и B, у которой число строк матрицы A равно числу
столбцов матрицы B.

ПРИМЕР 5. Даны две матрицы – матрица A раз-
мера (m×n) и матрица B размера (n×p), элементами кото-
рых являются вещественные числа. Описать процедуру,
строящую матрицу C размера (m×p), равную произведе-
нию матриц A и B.

Отметим,что все три матрицы имеют разные размеры,
поэтому в программе соответствующие им массивы будут
иметь разные описания. Описания типов при этом могут
выглядеть, например, так:

const mmax = 10; nmax = 15; pmax = 20;
type massa = array [1..mmax, 1..nmax] of real;

massb = array [1..nmax, 1..pmax] of real;
massc = array [1..mmax, 1..pmax] of real

Переменные, которые будут использоваться в качестве
фактических параметров при вызове процедуры, можно
описать так:

var a: massa;
b: massb;
c: massc

Приведем описание этой процедуры:

procedure Multiplication (a: massa; b: massb;
var c: massc);

var i, j, k: integer;
sum: real;

begin
for i := 1 to m do
for j := 1 to p do
begin
sum := 0;
for k := 1 to n do

sum := sum + a[i,k] * b[k,j];
c[i,j] := sum

end
end

Верхние границы параметров циклов m, n, p – это истин-
ные размеры матриц. Их значения должны быть опреде-
лены до вызова процедуры Multiplication.

ПРИМЕР 6. Дана матрица A размера (m×n). Опи-
сать функцию, вычисляющую ее вторую норму. Вторая
норма – это неотрицательное число, вычисляемое по фор-
муле

max
16j6n

m∑

i=1

|aij |:

Таким образом, чтобы найти вторую норму матрицы, нуж-
но просуммировать по столбцам значения ее элементов,
взятых по абсолютной величине, и из таких сумм выбрать

наибольшую. В программе описанию функции должно
предшествовать описание типа ее параметра:

const mmax = 10; nmax = 15;
type table = array [1..mmax, 1..nmax] of real

Приведем текст описания функции norm:

function norm (a: table): real;
var i, j: integer:

max, sum: real;
begin
max := 0;
for j := 1 to n do
begin
sum:=0;
for i := 1 to m do

sum := sum + a[i,j];
if sum > max then max := sum

end;
norm := max

end

ПРИМЕР 7. Написать программу, вычисляющую
сумму матричного ряда

S = eA = E +
A
1!

+
A2

2!
+ : : :+

An

n!
+ : : : ;

где A – квадратная матрица размера (m×m), с веществен-
ными элементами, E – единичная матрица, то есть мат-
рица, в которой элементы, стоящие на главной диагонали

равны единице, а остальные – нулю. Вычисление суммы
ряда закончить, как только норма очередного слагаемого
станет меньше заданного малого положительного числа
".

Написать эту программу будет совсем нетрудно, так
как в предыдущих примерах были разобраны почти все
нужные процедуры и функции: процедуры ввода и вы-
вода матриц, процедуры сложения и умножения матриц
и функция, вычисляющая норму матрицы.

Дополним этот список. Напомним, что умножить (раз-
делить) матрицу на число означает, что все ее элементы
должны быть умножены (разделены) на это число. Нуж-
ная нам процедура имеет описание

procedure Division (a:matrix; k: real; var b:matrix);
var i, j: integer;
begin
for i := 1 to m do
for j := 1 to m do

b[i,j] := a[i,j]/k
end

Опишем также процедуру, которая строит единичную мат-
рицу:

procedure UnityMatrix (var x: matrix);
var i, j: integer;
begin
for i := 1 to m do
for j := 1 to m do
if i=j then x[i,j] := 1 else x[i,j]:=0

end

С помощью следующей процедуры будет вводиться eps –
величина, характеризующая точность вычислений:

procedure Accurate (eps: real);
begin
writeln;
write (’введи точность ’);
readln (eps)

end

Пояснения к процедуре Series, вычисляющей резуль-
тирующую матрицу, сделаем после полного текста про-
граммы.

program SumOfSeries;
const mmax = 20;
type matrix = array [1..mmax, 1..mmax] of real;
var a, sum: matrix;

m: integer;
eps: real;

procedure SizeOfMatrix;
begin
writeln;
write (’введи размер матриц ’);
readln (m)

end;
procedure InputMatrix (var x: matrix);
var i, j: integer;
begin
writeln;
writeln (’Введи ’, m*m, ’ чисел’);

for i := 1 to m do
for j := 1 to m do

read (x[i,j])
end;

procedure OutputMatrix (x: matrix);
var i, j: integer;
begin
writeln;
writeln (’******* М А Т Р И Ц А *******’);
writeln;
for i := 1 to m do
begin
for j := 1 to m do
write (x[i,j]:8:7, ’ ’ );

writeln
end

end;
procedure Addition (x, y: matrix; var z: matrix);
var i, j: integer;
begin
for i := 1 to m do
for j := 1 to m do

z[i,j] := x[i,j] + y[i,j]
end;

procedure Division (x: matrix; k: integer;
var y: matrix);

var i, j: integer;
begin
for i := 1 to m do
for j := 1 to m do

y[i,j] := x[i,j]/k
end;

procedure Multiplication (a, b: matrix;
var c: matrix);

var i, j, k: integer;
sum: real;

begin
for i := 1 to m do
for j := 1 to m do
begin
sum := 0;
for k := 1 to m do
sum := sum + a[i,k] * b[k,j];

c[i,j] := sum
end

end;
procedure UnityMatrix (var x: matrix);
var i, j: integer;
begin
for i := 1 to m do
for j := 1 to m do
if i=j then x[i,j] := 1 else x[i,j]:=0

end;
function norm (x: matrix): real;
var i, j: integer;

max, sum: real;
begin
max := 0;
for j := 1 to m do
begin
sum:=0;

for i := 1 to m do
sum := sum + x[i,j];

if sum > max then max := sum
end;

norm := max
end;

procedure Accurate (eps: real);
begin
writeln;
write (’введи точность ’);
readln (eps)

end;
procedure Series(a: matrix; eps: real;

var sum: matrix);
var k: integer;

add: matrix;
begin
UnityMatrix (sum);
Multiplication (a, sum, add);
k:=1;
while norm (add) > eps do
begin

Addition(sum, add, sum);
Multiplication (add, a, add);
k:=k+1;
Division(add, k, add);

end
end;

begin
SizeOfMatrix;
InputMatrix (a);

OutputMatrix (a);
Accurate (eps);
Series (a, eps, sum);
OutputMatrix (sum)

end.

Поясним алгоритм, описанный в теле процедуры Series.
Вызов UnityMatrix(sum) определит начальное значе-
ние результирующей матрицы sum – сделает ее единич-
ной. С помощью вызова Multiplication(a, sum, add)
начальное значение, равное матрице a, получит матрица
add – очередное слагаемое. Целочисленная переменная
k, получив значение, равное 1, впоследствии будет при-
нимать значения 2, 3, 4, : : :

Оператор цикла, проверив норму очередного слагае-
мого, добавит его в результирующую матрицу sum ( вы-
зов Addition(sum, add, sum) ). Следующие три опе-
ратора вычисляют новую матрицу-слагаемое: предыду-
щее слагаемое домножается на матрицу a (так накапли-
вается степень), а увеличенное значение k используется в
вызове процедуры Division(add, k, add) – добавляется
очередной делитель. Все это произойдет, конечно, только
в том случае, если норма очередного слагаемого окажется
меньше числа eps.

Операторная часть программы представляет собой по-
следовательные обращения к процедурам и в коммента-
рии не нуждается.


